
Paradigmes de programmation
Atelier 4 : Modules et création de types – Haskell

I- PARTIE A : Les modules
Cette partie vise tout simplement à vous familiariser avec la manipulation de modules.
[bookmark: _GoBack]Créez le fichier Liste.hs et éditez-le pour faire le module Liste qui contient certaines fonctions que voici:
module Liste where

-- listeInverse :: [a] -> [a]
-- renvoie la liste dans l'ordre inverse
listeInverse [] = []
listeInverse (t:q) = listeInverse' q [t]
listeInverse' [] l = l
listeInverse' (t:q) l = listeInverse' q (t:l)

-- listeTriee :: Ord a => [a] -> [a]
-- renvoie la liste en ordre croissant
listeTriee [] = []
listeTriee l = listeTriee' l []
inserer e [] = [e]
inserer e (t:q)
    | e > t = t : inserer e q
    | otherwise = e:t:q
listeTriee' [] l = l
listeTriee' (t:q) l = listeTriee' q (inserer t l)

-- compteur :: (Eq a, Num b) => a -> [a] -> b
-- renvoie le nombre d'occurences d'un certain élément
compteur _ [] = 0
compteur e (t:q)
        | e == t = compteur e q + 1
        | otherwise = compteur e q
        
-- compteur2 :: Num a => (b -> Bool) -> [b] -> a
-- renvoie le nombre d'éléments qui répondent à un
-- certain critère
compteur2 _ [] = 0
compteur2 f (t:q)
    | f t = compteur2 f q + 1
    | otherwise = compteur2 f q

-- elem :: Eq a => a -> [a] -> Bool
-- renvoie True si l'élément fait partie de la liste
-- ou Flase sinon
elem _ [] = False
elem e (t:q)
    | e == t = True
    | otherwise = elem e q
Si vous avez nommé votre fonction elem comme ci-haut (plutôt qu'elem'), vous verrez l'erreur suivante en chargeant le fichier:
  Main> :l Liste.hs
[1 of 1] Compiling Liste            ( Liste.hs, interpreted )
Liste.hs:42:19:
    Ambiguous occurrence `elem'
    It could refer to either `Liste.elem', defined at Liste.hs:39:1
                          or `Prelude.elem',
                             imported from `Prelude' at Liste.hs:1:8-12
                             (and originally defined in `Data.Foldable')
Il y a plusieurs manières de remédier à ce problème. On pourrait donner un nom différent à notre fonction, mais admettons que nous tenons vraiment à elem. Il faut donc importer le module Prelude d'une manière qui nous permette de différentier les différentes versions d'elem.
Une première possibilité est d'importer Prelude en utilisant le mot-clé qualified pour que les noms importés soient qualifiés, c'est-à-dire que l'on doit utiliser le préfixe Prelude. . De cette manière, on différencie notre elem de Prelude.elem. Voyons voir ce qui se passe dans ce cas:
Rajouter la ligne suivante juste après le  « module Liste where » :
import qualified Prelude
Et lorsqu'on recharge le fichier...:
On obtient un gros tas d’erreurs…
Notre module ne reconnaît plus nos opérateurs (> …), qui étaient normalement importés de Prelude. Il faudrait, dans ce cas, remplacer chaque référence à ces fonctions par Prelude.lafonction.  
Mais ce n'est pas très pratique. Il existe donc un autre mécanisme en Haskell qui permet d'importer partiellement un module, c'est-à-dire en spécifiant de manière explicite certaines parties du module que nous ne voulons pas importer, en utilisant le mot-clé hiding suivi de la liste de ce que l'on veut cacher. Par exemple:
module Liste where
import Prelude hiding (elem)
De cette manière, il n'y a pas de conflit de nom entre les deux elem et tout le reste fonctionne sans Preluded'elem.
Maintenant que nous avons créé le module Liste, voyons comment nous pouvons l'utiliser dans nos programmes. Créez un nouveau fichier Haskell (nommé testListe.hs) et importez votre module Liste comme suit:
import Liste
elem2 :: Eq a => [a] -> Bool
elem2 (x:xs) = elem x xs
Remarquez ce qui se passe lorsque vous charger ce fichier:
Prelude> :l testListe.hs
[1 of 2] Compiling Liste            ( Liste.hs, interpreted )
[2 of 2] Compiling Main             ( testListe.hs, interpreted )
testListe.hs:4:16:
    Ambiguous occurrence `elem'
    It could refer to either `Prelude.elem',
                             imported from `Prelude' at testListe.hs:1:1
                             (and originally defined in `Data.Foldable')
                          or `Liste.elem',
                             imported from `Liste' at testListe.hs:1:1-12
                             (and originally defined at Liste.hs:41:1-4)
Failed, modules loaded: Liste.
Puisque Prelude est importé par défaut, on aura toujours ce problème si on importe Liste sans rien spécifier d'autre. On pourrait utiliser le mot-clé hiding pour cacher l'une des deux définitions, ou importer Liste de manière qualifiée. Par exemple:
import Liste hiding(elem)
-- ou --
import Prelude hiding(elem)
import Liste
-- ou --
import qualified Liste
Une autre possibilité est d'importer seulement les fonctions dont nous avons besoin. Par exemple, si on veut seulement importer listeTriee et compteur2, on peut écrire:
import Liste (listeTriee, compteur2)

nbPlusGrandsQue5 = compteur2 (>5)
nombres = [1,432,7,787,678,2,12,678,965,3,234,56,87]
nombresEnOrdre = listeTriee nombres
On peut aussi combiner le mot-clé qualified avec hiding pour importer qu'une partie des fonctions et forcer l'écriture de leur nom complet:
import qualified Liste (listeTriee, compteur2)
nbPlusGrandsQue5 = Liste.compteur2 (>5)
Remarquez que si on importe Liste sans spécifier ce qui est importé (avec qualified ou hiding), on a toujours accès à des fonctions privé qui ne sont utiles que localement telles que listeInverse' et listeTriee', et qui ne devraient pas normalement être visibles de l'extérieur du module. Par exemple:
Main> Liste.listeTriee' [2,5,3] [2,5,3]
[2,2,3,5,5,3]
On peut spécifier les éléments qui peuvent être exportées du module Liste, de manière à ce que ceux qui l'importent n'aient accès qu'à certaines fonctions, certains types, certaines variables, etc. Dans la déclaration du module, on peut inclure la liste de ce qui doit être exporté après le nom du module. Par exemple:
module Liste (listeInverse, listeTriee, compteur, compteur2, elem) where
import Prelude hiding (elem)
Notez que l'on peut aussi spécifier des types à exporter lors de la création d'un module. Par exemple:
module Transports (Vehicule(..), nbCamions, constructeurs)
data Vehicule =
    Voiture String String String String
    | Camion String String Int String
    | Moto String String String
    deriving (Show)
etc.
Dans ce cas, le (..) signifie que l'on exporte tous les constructeurs de ce type. On peut importer un type de la même manière:
import Transports (Vehicule(..))
-- ou --
import Transports hiding (Vehicule(..))
On peut aussi spécifier qu'un module exporte certaines fonctionnalités d'un autre module. Par exemple, si on se sert du module Transports pour manipuler des listes de véhicules et on veut donner accès aux fonctions de Liste, on peut importer Liste dans Transports et exporter le module complet:
module Transports (Vehicule(..), nbCamions, constructeurs, module Liste) where  
import Liste hiding(elem)
Et ensuite:
import Transports hiding(elem)
nbPlusGrandsQue5 = compteur2 (>5)

II- PARTIE B : Création de types
1.  type Personne
Définissez les types Nom, Prenom et NAS de la manière suivante avec le mot-clé type:
type Nom = String
type Prenom = String
type NAS = Int
Définissez quelques noms et prénoms. Par exemple:
p1= "Jacques" :: Prenom
p2= "Jean" :: Prenom
p3= "Sophie" :: Prenom
n1= "Robert" :: Nom
n2= "Jean" :: Nom
etc.
Remarquez ensuite que les types Nom et Prenom sont parfaitement équivalents à String (ou [Char]). Par exemple:
Main> p1
"Jacques"
Main> p1 == p2
False
Main> p1 == p1
True
Main> p1 == n1
False
Main> p2 == n2
True
Main> p1 > p2
False
Main> p1 > p3
False
Main> n1 == "Robert"
True
Main> :type n2
n2 :: [Char]
Main> reverse n1
"treboR"
etc.
Définissez le type personne à l'aide du mot-clé type et des types Nom, Prenom et NAS, puis créér quelques personnes. Par exemple:
type Personne= (Nom, Prenom, NAS)

pers1= (n1, p1, 987654321) :: Personne
pers2= (n1, p2, 987123456) :: Personne
pers3= (n2, p1, 456789321) :: Personne
pers4= (n2, p3, 789654123) :: Personne
Vous pouvez remarquer que ce type est aussi équivalent à ([Char], [Char], Int). Par exemple:
Main> pers3 == ("Jean", p1, 456789321)
True
Main> pers3 < pers2
True
Main> pers2
("Robert","Jean",987123456)
etc.
Crééz ensuite les fonctions suivantes pour extraire certaines données d'une personne:
nom :: Personne -> Nom
prenom :: Personne -> Prenom
nas :: Personne -> NAS

Par exemple:
Main> nom pers1
"Robert"
Main> nas pers1
987654321
Main> nom ("Latour", "Gilles", 546789321)
"Latour"
Main> :type nom
nom :: Personne -> Nom
etc.
Créez ensuite une liste de personnes et triez-la. Vous pouvez utiliser la fonction listeTriee de l’atelier 2. 
Par exemple:
personnes= [pers1, pers2, ("Latour", "Gilles", 546789321), pers3, pers4]

Main> personnes
[("Tremblay","Jean",987654321),("Tremblay","Jacques",987123456),("Latour","Gilles",546789321),("Jean","Jean",456789321),("Jean","Julien",789654123)]
Main> listeTriee personnes
[("Jean","Jacques",456789321),("Jean","Sophie",789654123),("Latour","Gilles",546789321),("Robert","Jacques",987654321),("Robert","Jean",987123456)]
Main> :type personnes
personnes :: [Personne]
etc.
2.  type Etudiant
Définissez le type CodePerm comme étant équivalent à String, puis définissez le type Etudiant comme l'équivalent de (Nom, Prenom, CodePerm). Crééz ensuite quelques étudiants et faites quelques tests:
Main> etu1
("Lapointe","Marc","LM123452")
Main> etu1==etu2
False
Main> etu1>etu2
False
Main> etu1<etu2
True
Main> nom etu1
       <interactive>:75:5:
               Couldn't match type `[Char]' with `Int'
               Expected type: Personne
                      Actual type: Etudiant
               In the first argument of `nom', namely `etu1'
               In the expression: nom etu1
etc.

... et si on avait un type Etudiant', avec le code permanent numérique...
type CodePermNum= Int
type Etudiant'= (Nom, Prenom, CodePermNum)

etu'1= (n1, p1, 456321444) :: Etudiant'
etu'2= (n2, p3, 789654123) :: Etudiant'
...
Main> nom etu'1
"Robert"
Main> etu'2 == pers4
True      ---hummmm !!!!
Main> etu'2 == etu2
<interactive>:85:10:
    Couldn't match type `[Char]' with `Int'
    Expected type: Etudiant'
      Actual type: Etudiant
    In the second argument of `(==)', namely `etu2'
    In the expression: etu'2 == etu2
Main> nas etu'1
456321444               -- faux!
etc.

3.  nouveau type Automobile
Définissez le type Automobile de la manière suivante avec le mot-clé newtype:
-- Auto (constructeur, modele, couleur, immatriculation)
newtype Automobile= Auto (String, String, String, String)
Crééz ensuite quelques autos:
auto1= Auto ("Ford", "Festiva", "rouge", "QWE 321")
auto2= Auto ("Toyota", "Camry", "bleu", "NLR 606")
auto3= Auto ("Ford", "Mustang", "jaune", "DVL 666")
Essayez maintenant quelques opérations sur des autos. Remarquez que ce type n'est pas équivalent à (String, String, String, String):
Main> auto1
<interactive>:103:1:
    No instance for (Show Automobile) arising from a use of `print'
    In a stmt of an interactive GHCi command: print it
Main> auto1 == auto1
<interactive>:105:7:
    No instance for (Eq Automobile) arising from a use of `=='
    In the expression: auto1 == auto1
    In an equation for `it': it = auto1 == auto1
Main> auto1 == ("Ford", "Festiva", "rouge", "QWE 321")
<interactive>:110:10:
    Couldn't match expected type `Automobile'
                with actual type `([Char], [Char], [Char], [Char])'
    In the second argument of `(==)', namely
      `("Ford", "Festiva", "rouge", "QWE 321")'
    In the expression: auto1 == ("Ford", "Festiva", "rouge", "QWE 321")
    In an equation for `it':
        it = auto1 == ("Ford", "Festiva", "rouge", "QWE 321")
etc.
Créez quelques fonctions comme:
constructeur :: Automobile -> String
modele :: Automobile -> String
couleur :: Automobile -> String
immatriculation :: Automobile -> String
Main> constructeur auto1
"Ford"
Main> modele auto2
"Camry"
Main> couleur auto3
"jaune"
Main> :type immatriculation
immatriculation :: Automobile -> String
etc.
Modifiez votre type Automobile et dites à Haskell qu'il dérive des classes Eq et Show:
newtype Automobile = Auto (String, String, String, String) deriving (Eq, Show)

Main> auto1
Auto ("Ford","Festiva","rouge","QWE 321")
Main> auto1 == auto1
True
Main> auto1 == ("Ford", "Festiva", "rouge", "QWE 321")
<interactive>:110:10:
    Couldn't match expected type `Automobile'
                with actual type `([Char], [Char], [Char], [Char])'
    In the second argument of `(==)', namely
      `("Ford", "Festiva", "rouge", "QWE 321")'
    In the expression: auto1 == ("Ford", "Festiva", "rouge", "QWE 321")
    In an equation for `it':
        it = auto1 == ("Ford", "Festiva", "rouge", "QWE 321")
Main> auto1 == Auto("Ford", "Festiva", "rouge", "QWE 321")
True
Main> auto1 < auto2
<interactive>:133:7:
    No instance for (Ord Automobile) arising from a use of `<'
    In the expression: auto1 < auto2
    In an equation for `it': it = auto1 < auto2
On voit qu'on a maintenant des fonctions (==) et show pour Automobile, qui sont basées sur celles des tuples, des listes et des String. On n'a cependant pas de fonctions d'ordre (< > >= etc.). On peut définir nous-mêmes la manière dont Haskell compare deux Automobiles, en lui disant qu'automobile est une instance de Ord et en définissant la fonction compare:
instance Ord Automobile where
                        compare x y
                            | immatriculation x == immatriculation y = EQ
                            | immatriculation x < immatriculation y = LT
                            | otherwise = GT

Main> auto1 < auto2
False
Main> auto1 > auto2
True
Main> listeTriee [auto1, auto2, auto3]
[Auto ("Ford","Mustang","jaune","DVL 666"),Auto ("Toyota","Camry","bleu","NLR 6 
06"),Auto ("Ford","Festiva","rouge","QWE 321")]

4. Type de données Vehicule
Définissez le type Vehicule avec le mot-clé data. Il peut y avoir trois types de véhicules:
Voiture [constructeur] [modèle] [couleur] [immatriculation]
Camion [constructeur] [modèle] [nb.essieux] [immatriculation]
Moto [constructeur] [modèle] [immatriculation]
On veut pouvoir montrer un véhicule avec la fonction show par défaut. Par exemple:
v1= Voiture "Ford" "Festiva" "gris" "ASD 987"
v2= Camion "GMC" "12345" 3 "FGH 123"
v3= Moto "Honda" "X600" "ZXC 321"

Main> v1
Voiture "Ford" "Festiva" "gris" "ASD 987"
Main> v2
Camion "GMC" "12345" 3 "FGH 123"
Main> v3
Moto "Honda" "X600" "ZXC 321"
Main> :type v1
v1 :: Vehicule
Main> v1 > v2
<interactive>:171:4:
    No instance for (Ord Vehicule) arising from a use of `>'
    In the expression: v1 > v2
    In an equation for `it': it = v1 > v2

Définissez les fonctions suivantes pour extraire des données d'un véhicule:
constructeur' :: Vehicule -> [Char]
modele' :: Vehicule -> [Char]
immatriculation' :: Vehicule -> [Char]

Main> constructeur' v2
"GMC"
Main> constructeur' (Voiture "Honda" "Prelude" "vert" "ZZZ 999")
"Honda"
Main> :type constructeur'
constructeur' :: Vehicule -> [Char]

Définissez les fonctions nécessaires (et de la bonne manière) pour qu'on puisse comparer deux véhicules (égalité, ordre) selon leur immatriculation. On devrait ensuite pouvoir faire:
Main> v1 == v2
False
Main> v1 > v2
False
Main> v1 < v2
True
Main> listeTriee [v1, v2, v3, Voiture "Pontiac" "Sunfire" "blanc" "IJN 345"]
[Voiture "Ford" "Festiva" "gris" "ASD 987",Camion "GMC" "12345" 3 "FGH 123",Voiture "Pontiac" "Sunfire" "blanc" "IJN 345",Moto "Honda" "X600" "ZXC 321"]

Définissez la fonction nbCamions qui renvoie le nombre de camions dans une liste de véhicules. Par exemple:
Main> nbCamions [v1, v2, v3, Voiture "Pontiac" "Sunfire" "blanc" "IJN 345"]
1
Main> nbCamions [v1, v3, v1, v3]
0
Main> nbCamions [v1, v2, v3, v2, v2]
3
Main> :type nbCamions
nbCamions :: Num a => [Vehicule] -> a
Définissez la fonction constructeurs qui renvoie une liste de constructeurs à partir d'une liste de véhicules. 
Main> constructeurs [v1, v2, v3, Voiture "Pontiac" "Sunfire" "blanc" "IJN 345"]
["Ford","GMC","Honda","Pontiac"]
Main> constructeurs [v1, v2, v3, v2, v2]
["Ford","GMC","Honda","GMC","GMC"]
Main> :type constructeurs
constructeurs :: [Vehicule] -> [String]
Définissez la fonction vehiculesDuConstructeur qui renvoie une liste des véhicules d'un certain constructeur à partir d'une liste de véhicules. 
Main> vehiculesDuConstructeur [v1, v2, v3] "GMC"
[Camion "GMC" "12345" 3 "FGH 123"]
Main> vehiculesDuConstructeur [v1, v2, v3, Voiture "Ford" "Mustang" "jaune" "DVL 666"] "Ford"
[Voiture "Ford" "Festiva" "gris" "ASD 987",Voiture "Ford" "Mustang" "jaune" "DVL 666"]
Main> vehiculesDuConstructeur [v1, v2, v3] "Toyota"
[]
Main> :type vehiculesDuConstructeur
vehiculesDuConstructeur :: [Vehicule] -> String -> [Vehicule]

